RESEARCH REPORT

NO. 118

DETERMINING TRANSPORT PARAMETERS FROM SOLUTE DISPLACEMENT EXPERIMENTS

MARTINUS TH. VAN GENUCHTEN

APRIL 1980

UNITED STATES DEPARTMENT OF AGRICULTURE
SCIENCE AND EDUCATION ADMINISTRATION
U.S. SALINITY LABORATORY
RIVERSIDE, CALIFORNIA

TABLE OF CONTENTS

DETERMINING TRANSPORT PARAMETERS FROM SOLUTE DISPLACEMENT EXPERIMENTS[1/]

Martinus Th. van Genuchten[2/]

ABSTRACT

Predictions of solute transport in the field are generally based upon convective-dispersive type transport equations. The one-dimensional form of this equation contains two parameters which must be determined beforehand. They are the dispersion coefficient and a distribution coefficient, the latter accounting for adsorption or exchange between liquid and solid phases. Both coefficients can be obtained by fitting an analytical solution of the one-dimensional convective-dispersive transport equation to observed column effluent data. This paper describes a non-linear least-squares curve-fitting computer model which may be used for that purpose. A listing of the program is given in an appendix.

1/ Research Report, U. S. Salinity Laboratory, USDA-SEA-AR, Riverside, California, and Department of Soil and Environmental Sciences, University of California, Riverside.

2/ Present address: U. S. Salinity Laboratory, 4500 Glenwood Drive, Riverside, California 92501.

INTRODUCTION

Predictions of solute transport in the field are often obtained by solving convective-dispersive type transport equations. For a one-dimensional system at constant water content and steady-state flow, the appropriate equation is

$$R \frac{\partial c}{\partial t} = D \frac{\partial^2 c}{\partial x^2} - v \frac{\partial c}{\partial x} \qquad [1]$$

where

c = solute concentration,
D = dispersion coefficient,
R = retardation factor
t = time,
v = average pore-water velocity,
x = distance.

The parameter R in Eq. [1] accounts for possible interactions between the chemical and the solid phase of the soil. For the special case of chemical adsorption or exchange, the retardation factor is given by (van Genuchten et al., 1974)

$$R = 1 + \frac{\rho k}{\theta} \qquad [2]$$

where

k = distribution coefficient,
θ = volumetric water content,
ρ = soil bulk density.

Equation [2] assumes that the adsorption or exchange reactions are always instantaneous (equilibrium adsorption). If there is no interaction between the chemical and the solid phase, k in Eq. [2] becomes zero and R reduces to one. In some cases R may become less than one, indicating that only a fraction of the liquid phase participates in the transport process. This occurs when the tracer is subject to anion exclusion (e.g., for chloride movement in fine-textured soils), or when immobile liquid regions are present which do not contribute to convective solute transport (water in dead-end pores or inside dense aggregates). In the case of anion exclusion, (1-R) can be viewed as the relative anion exclusion volume, and (-k) as the specific anion exclusion volume (e.g., expressed in cm^3 water per gram of soil).

Equation [1] contains two parameters (R and D) which need to be quantified before the equation can be used for actual prediction purposes. Estimates for these two parameters are often obtained by analyzing experimental column effluent curves. Several methods of analysis are available for that purpose. Rifai et al. (1956), for example, proposed a method for calculating D from the slope of a breakthrough curve. Rose and Passioura (1971) and Passioura et al. (1970) discuss a procedure which allows D and R to be determined from a plot of ln(t) versus c on probability paper. Agneessens et al. (1978) used the method of moments to obtain D from pulse-type effluent curves. Another method, probably the most accurate one, is based upon a least-squares analysis of the effluent data (Elprince and Day, 1977; Laudelout and Dufey, 1977; Agneessens et al., 1978; Le Renard, 1979). For that purpose an appropriate analytical solution of Eq. [1] is fitted to the effluent data, thereby allowing D and R to be estimated simultaneously from the experimental data. This method is further discussed in this paper.

REVIEW OF ANALYTICAL SOLUTIONS

Several analytical solutions of Eq. [1] exist, each one based upon a different set of initial and boundary conditions. Appendix A lists five analytical solutions which are the most useful for the purpose of analyzing effluent data. It is noted here that each set of boundary conditions involves different assumptions regarding the physics of the column displacement experiment. This paper does not deal with the advantages and disadvantages of each boundary condition, but rather will use each analytical solution as a basis for estimating R and D from observed effluent data.

The different analytical solutions in Appendix A are expressed in terms of the original variables (x, t, R, v, D). For an analysis of the effluent data it is more convenient to introduce the dimensionless variable

$$T = vt/L, \qquad [3]$$

and the dimensionless group

$$P = vL/D, \qquad [4]$$

where T is the number of pore volumes, P is the column Peclet number, and L is the column length. The effluent concentrations at x=L, $c_e(T)$, can then be represented by much simpler equations. Appendix B gives expressions for the effluent concentration for each of the five analytical solutions given in Appendix A.

The analytical solutions in Appendices A and B are applicable only to column experiments where the tracer is applied continuously at the inlet position (continuous tracer application). For pulse-type applications, the expressions must be replaced by

$$c^*(x,t) = \begin{cases} c(x,t) & 0 < t \leq t_1 \\ c(x,t) - c(x,t-t_1) & t > t_1 \end{cases} \qquad [5]$$

$$c_e^*(T) = \begin{cases} c_e(T) & 0 < T \leq T_1 \\ c_e(T) - c_e(T-T_1) & T > T_1 \end{cases} \qquad [6]$$

where

t_1 = time length of tracer pulse added to column,
T_1 = vt_1/L,
$c^*(x,t)$ = concentration for pulse-type application.
$c_e^*(T)$ = effluent concentration for pulse-type application.

DETERMINING P AND R FROM THE EFFLUENT CURVE

A computer program was written which allows one to fit any of the five analytical solutions for c_e (or c_e^*) to observed effluent data. The program is a simplification of the multi-purpose, non-linear, least-squares program of Meeter (1964). The curve-fitting technique uses the maximum neighborhood method of Marquardt (1963), which is based upon an optimum interpolation between Taylor series expansions and the method of steepest descent. A detailed description of this particular technique is given in Daniel and Wood (1973).

In the case of a continuous tracer application (i.e., for c_e), only two parameters (P and R) need to be determined. When a pulse-type tracer effluent curve is present (c_e^*), information is also needed about the dimensionless pulse length, T_1. Although T_1 is often available from the experimental conditions, it is often more convenient (and accurate) to also estimate this parameter from the experimental curve. The computer program therefore allows both a two-parameter (P and R, P and T_1, or R and T_1) or a three-parameter (P, R, and T_1) curve-fitting to be carried out. Appendix C gives a short description and listing of the program.

APPLICATIONS

The accuracy of the curve-fitting program was first tested by fitting P and R to three hypothetical effluent curves with known values of P and R. The analytical solution of Case SI-1 in Appendix B (Eq. [B2] was used to generate these "observed" curves. Calculated curves were obtained for P = 10, 40, and 400, and R = 1. The datapoints in each case were distributed in equal intervals along the T-axis, and located between relative concentrations of 0.05 and 0.95 (34 datapoints for P=10, 19 points for P=40, and 12 points for P=400). Table 1 gives the curve-fitted values of P and R for all five analytical solutions. As expected, the input values of P and R were duplicated exactly when the analytical solution of Case SI-1 was fitted to the "observed" effluent curve. The curve-fitted values of P and R for the other analytical solutions, however, deviate from the input values. This, of course,

Table 1. Curve-fitted values of P and R for the five analytical solutions in Appendix B. The "observed" effluent curves were based on Eq. [B2], with R=1, and P = 10, 40, and 400, respectively.

Input value of P (SI-1)	Case INF	Case SI-1	Case SI-2	Case FN-1	Case FN-2
	FITTED VALUE OF P				
10.00	10.46	10.00	9.58	9.11	8.92
40.00	40.49	40.00	39.52	39.40	38.96
400.00	400.45	399.96	399.46	399.45	400.00
	FITTED VALUE OF R				
10.00	0.911	1.000	0.904	1.124	0.999
40.00	0.976	1.000	0.975	1.026	1.000
400.00	0.998	1.000	0.998	1.003	0.999

is to be expected since each analytical solution is based upon different boundary conditions. Differences between fitted and input values are greatest when P is small, i.e., for short soil columns (Table 1).

A second example considers the movement of Chromium through sand (Wierenga, 1980; unpublished data). Observed effluent data from the 5-cm long soil column are shown in Fig. 1. The analytical solutions of Cases SI-2 and FN-2 were fitted to these data. Results of the two-parameter curve-fittings are given in Table 2. The solid line in Fig. 1 represents the fitted analytical solution for Case SI-2. The fitted curve for FN-2 was found to be essentially the same as for SI-2, even though the estimated parameters P and R are different (Table 2). By making use of Eq. [4] and [2], it is possible to obtain also estimates for the dispersion coefficient (D) and the distribution constant (k). Table 2 shows that the values of D and k are somewhat different for the two analytical models. The differences between the estimated values could have been made smaller if a longer soil column were used for the displacement experiment (see also Table 1).

A third example considers the movement of a pulse of Chloride through Norge Loam (Davidson, 1973; unpublished data). Figure 2 com-



Fig. 1. Observed and curve-fitted breakthrough curves for Chromium movement through sand (example 2).

pares the observed and fitted effluent curves (Case SI-2). The fitted curve for Case SI-1 was again essentially the same as for Case FN-2. In this case all three parameters (P, R, and T_1) were fitted to the data (Table 2). Note that the estimated values of P are much higher than for the previous experiment. The value of R is less than 1, indicating some anion exclusion. The specific anion exclusion volume (-k) is about 0.02 cm^3 water per gram of soil (Table 2).

Table 2. Measured and curve-fitted parameters for two column displacement experiments.

Experiment	2A	2B	3A	3B
Tracer	Cr^{6+}	Cr^{6+}	Cl^-	Cl^-
Model	SI-2	FN-2	SI-2	FN-2
θ (cm^3/cm^3)	0.184	0.184	0.3626	0.3626
ρ (g/cm^3)	1.679	1.679	1.527	1.527
v (cm/day)	19.64	19.64	14.23	14.23
L (cm)	5.0	5.0	30.0	30.0
P (fitted)	19.19	18.59	287.4	287.4
R (fitted)	1.281	1.349	0.918	0.918
T_1 (fitted)	-	-	0.408	0.408
T_1 (measured)	-	-	0.425	0.425
D (cm^2/day)	5.12	5.28	1.49	1.49
k (cm^3/g)	0.031	0.038	-0.019	-0.019



Fig. 2. Observed and curve-fitted effluent curves for Chloride movement through Norge loam (example 3).

CONCLUSIONS

The least-squares computer model discussed in this report provides an easy to use, efficient and accurate means of fitting various transport parameters to observed column effluent data. The unknown parameters include the column Peclet number, P, the retardation factor, R, and the dimensionless pulse time, T_1. The three examples, furthermore, demonstrate that the use of different analytical solutions can lead to large differences between the curve-fitted P- and R-values. These differences are most significant when P is small, i.e., for relatively short soil columns.

REFERENCES

Agneessens, J.-P., P. Dreze, and L. Sine. 1978. Modélisation de la migration d'éléments dans les sols. II. Détermination du coefficient de dispersion et de la porosité efficace. Pedologie, 27: 373-388.

Brenner, H. 1962. The diffusion model of longitudinal mixing in beds of finite length. Numerical values. Chem. Eng. Sci. 17: 229-243.

Cleary, R. W., and D. D. Adrian. 1973. Analytical solution of the convective dispersive equation for cation adsorption in soils. Soil Sci. Soc. Am. Proc. 37: 197-199.

Danckwerts, P. V. 1953. Continuous flow systems. Chem. Eng. Sci. 2: 1-13.

Daniel, C., and F. S. Wood. 1973. Fitting Equations to Data. Wiley-Interscience, New York. 350 p.

Elprince, A. M., and P. R. Day. 1977. Fitting solute breakthrough equations to data using two adjustable parameters. Soil Sci. Soc. Am. J.41: 39-41.

Laudelout, H., and J. E. Duffy. 1977. Analyse numérique des expériences de lixiviation en sol homogène. Ann. Agron. 28: 65-73.

Lapidus, L., and N. R. Amundson. 1952. Mathematics of adsorption in beds. IV. The effect of longitudinal diffusion in ion exchange chromatographic columns. J. Phys. Chem. 56: 984-988.

Le Renard, J. 1979. Description d'une méthode de détermination du coefficient de dispersion hydrodynamique longitudinal en milieu poreux homogène saturé en eau. II. Utilisation d'un modele mathematique à trois paramètres dont un phénoménologique. Ann. Agron. 30: 109-119.

Lindstrom, F. T., R. Haque, V. H. Freed, and L. Boersma. 1967. Theory on the movement of some herbicides in soils: linear diffusion and convection of chemicals in soils. Environ. Sci. Technol. 1: 561-565.

Marquardt, D. W. 1963. An algorithm for least-squares estimation of non-linear parameters. J. Soc. Ind. Appl. Math. 11: 431-441.

Meeter, D. A. 1964. Non-linear least-squares (Gaushaus). Univ. of Wisconsin Computing Center. Program revised 1966.

Passioura, J. B., D. A. Rose, and K. Haszler. 1970. Lognorm: a program for analysing experiments on hydrodynamic dispersion. Technical Memorandum 70/6, CSIRO, Div. of Land Res., Canberra.

Rifai, M. N. E., W. J. Kaufman, and D. K. Todd. 1956. Dispersion phenomena in laminar flow through porous media. Sanitary Eng. Res. Lab., Univ. of Calif., Berkeley. Inst. of Eng. Res. Ser. No. 93(2).

Rose, D. A., and J. B. Passioura. 1971. The analysis of experiments on hydrodynamic dispersion. Soil Sci. 111: 252-257.

van Genuchten, M.Th., J.M. Davidson, and P.J. Wierenga. 1974. An evaluation of kinetic and equilibrium equations for the prediction of pesticide movement through porous media. Soil Sci. Soc. Am. Proc. 38: 29-35.

van Genuchten, M. Th., and W. J. Alves. 1980. A compendium of available analytical solutions of the one-dimensional convective-dispersive transport equation. U. S. Salinity Laboratory, Riverside, California.

APPENDIX A. Analytical solutions of the one-dimensional convective-dispersive transport equation (Eq. [1] for different initial and boundary conditions.

Case INF (Infinite System).

The solution of Eq. [1], subject to

$$c(x,0) = \begin{cases} 1 & x < 0 \\ 1/2 & x = 0 \\ 0 & x > 0 \end{cases} \quad \text{[A1]}$$

$$c(-\infty,t) = c_o \quad \text{[A2]}$$

$$c(\infty,t) = 0 \quad \text{[A3]}$$

is (Danckwerts, 1953)

$$c/c_o = \frac{1}{2}\,\mathrm{erfc}\left[\frac{Rx - vt}{2(DRt)^{1/2}}\right] \quad \text{[A4]}$$

where erfc is the complementary error function.

Case SI-1 (Semi-infinite system, first-type boundary condition).

The solution of Eq. [1], subject to

$$c(x,0) = 0 \qquad (x > 0) \quad \text{[A5]}$$

$$c(0,t) = c_o \quad \text{[A6]}$$

$$\frac{\partial c}{\partial x}(\infty,t) = 0 \quad \text{[A7]}$$

is (Lapidus and Amundson, 1952)

$$c/c_o = \frac{1}{2}\,\mathrm{erfc}\left[\frac{Rx - vt}{2(DRt)^{1/2}}\right] + \frac{1}{2}\exp(vx/D)\,\mathrm{erfc}\left[\frac{Rx + vt}{2(DRt)^{1/2}}\right] \qquad [A8]$$

Case SI-2 (Semi-infinite system, third-type boundary condition).

The solution of Eq. [1], subject to

$$c(x,0) = 0 \qquad (x > 0) \qquad [A9]$$

$$\left(-D\frac{\partial c}{\partial x} - vc\right)\Big|_{x=0} = vc_o \qquad [A10]$$

$$\frac{\partial c}{\partial x}(\infty,t) = 0 \qquad [A11]$$

is (Lindstrom et al., 1967)

$$c/c_o = \frac{1}{2}\,\mathrm{erfc}\left[\frac{Rx - vt}{2(DRt)^{1/2}}\right] + \left(\frac{v^2t}{\pi DR}\right)^{1/2}\exp\left[-\frac{(Rx - vt)^2}{4DRt}\right]$$

$$-\frac{1}{2}\left(1 + \frac{vx}{D} + \frac{v^2t}{DR}\right)\exp(vx/D)\,\mathrm{erfc}\left[\frac{Rx + vt}{2(DRt)^{1/2}}\right]. \qquad [A12]$$

Case FN-1 (Finite profile, first-type boundary condition).

The solution of Eq. [1], subject to

$$c(x,0) = 0 \qquad (x > 0) \qquad [A13]$$

$$c(0,t) = c_o \qquad [A14]$$

$$\frac{\partial c}{\partial x}(L,t) = 0 \qquad [A15]$$

is (Cleary and Adrian, 1973)

$$c/c_o = 1 - \sum_{m=1}^{\infty} \frac{2\beta_m \sin\left(\frac{\beta_m x}{L}\right) \exp\left[\frac{vx}{2D} - \frac{v^2 t}{4DR} - \frac{\beta_m^2 Dt}{L^2 R}\right]}{\beta_m^2 + \left(\frac{vL}{2D}\right)^2 + \frac{vL}{2D}} \qquad [A16]$$

where the eigenvalues β_m are the positive roots of

$$\beta_m \cot(\beta_m) + \frac{vL}{2D} = 0. \qquad [A17]$$

The series solution converges very slowly for large values of (vL/D) and/or small values of (vt/RL). For

$$\frac{vL}{D} > 5 + 40 \frac{vt}{RL} \qquad [A18]$$

or

$$\frac{vL}{D} > 100 \qquad [A19]$$

the following approximation gives very accurate answers (van Genuchten and Alves, 1980)

$$c/c_o = \frac{1}{2} \operatorname{erfc}\left[\frac{Rx - vt}{2(DRt)^{1/2}}\right] + \frac{1}{2} \exp(vx/D) \operatorname{erfc}\left[\frac{Rx + vt}{2(DRt)^{1/2}}\right]$$

$$+ \frac{1}{2} \left[2 + \frac{v(2L - x)}{D} + \frac{v^2 t}{DR}\right] \exp(vL/D) \operatorname{erfc}\left[\frac{R(2L-x) + vt}{2(DRt)^{1/2}}\right]$$

$$- \left(\frac{v^2 t}{\pi DR}\right)^{1/2} \exp\left[\frac{vL}{D} - \frac{R}{4Dt}\left(2L-x + \frac{vt}{R}\right)^2\right]. \qquad [A20]$$

Case FN-2 (Finite profile, third-type boundary condition).

The solution of Eq. [1], subject to

$$c(x,0) = 0 \qquad (x > 0) \qquad [A21]$$

$$\left(- D \frac{\partial c}{\partial x} + vc\right)\Big|_{x=0} = vc_o \qquad [A22]$$

$$\frac{\partial c}{\partial x}(L,t) = 0 \qquad [A23]$$

is (Brenner, 1962)

$$c/c_o = 1 - \sum_{m=1}^{\infty} \frac{\frac{2vL}{D}\beta_m[\beta_m \cos(\frac{\beta_m x}{L}) + \frac{vL}{2D}\sin(\frac{\beta_m x}{L})\exp[\frac{vx}{2D} - \frac{v^2 t}{4DR} - \frac{\beta_m^2 Dt}{L^2 R}]}{[\beta_m^2 + (\frac{vL}{2D})^2 + \frac{vL}{D}][\beta_m^2 + (\frac{vL}{2D})^2]} \qquad [A24]$$

where the eigenvalues of β_m are the positive roots of

$$\beta_m \cot(\beta_m) - \frac{\beta_m^2 D}{vL} + \frac{vL}{4D} = 0 \qquad [A25]$$

Also this series solution converges slowly for large values of (vL/D) and/or small values of (vt/RL). For conditions [A18,19] the following approximate solution provides accurate answers (Brenner, 1962):

$$c/c_o = \frac{1}{2}\,\mathrm{erfc}\left[\frac{Rx - vt}{2(DRt)^{1/2}}\right] + \left(\frac{v^2 t}{\pi DR}\right)^{1/2} \exp\left[- \frac{(Rx - vt)^2}{4DRt}\right]$$

$$- \frac{1}{2}\left(1 + \frac{vx}{D} + \frac{v^2 t}{DR}\right) \exp(vx/D)\,\mathrm{erfc}\left[\frac{Rx + vt}{2(DRt)^{1/2}}\right]$$

$$+ \left(\frac{4v^2 t}{\pi DR}\right)^{1/2} \left[1 + \frac{v}{4D}\left(2L-x + \frac{vt}{R}\right)\right] \exp\left[\frac{vL}{D} - \frac{R}{4Dt}\left(2L-x + \frac{vt}{R}\right)^2\right]$$

$$- \frac{v}{D}\left[2L-x + \frac{3vt}{2R} + \frac{v}{4D}\left(2L-x + \frac{vt}{R}\right)^2\right] \exp(vL/D)\,\mathrm{erfc}\left[\frac{R(2L-x) + vt}{2(DRt)^{1/2}}\right]$$

[A26]

APPENDIX B. Effluent concentration, $c_e(T)$, for each of the five analytical solutions given in Appendix A.

Case INF (Infinite system)

$$c_e/c_o = \frac{1}{2}\,\mathrm{erfc}\,[(\frac{P}{4RT})^{1/2}(R-T)] \qquad [B1]$$

Case SI-1 (Semi-infinite system, first-type boundary condition).

$$c_e/c_o = \frac{1}{2}\,\mathrm{erfc}[(\frac{P}{4RT})^{1/2}(R-T)] + \frac{1}{2}\exp(P)\,\mathrm{erfc}[(\frac{P}{4RT})^{1/2}(R+T)] \qquad [B2]$$

Case SI-2 (Semi-infinite system, third-type boundary condition).

$$c_e/c_o = \frac{1}{2}\,\mathrm{erfc}[(\frac{P}{4RT})^{1/2}(R-T)] + (\frac{PT}{\pi R})^{1/2}\exp[-\frac{P}{4RT}(R-T)^2]$$

$$-\frac{1}{2}(1 + P + \frac{PT}{R})\exp(P)\,\mathrm{erfc}[(\frac{P}{4RT})^{1/2}(R+T)] \qquad [B3]$$

Case FN-1 (Finite system, first-type boundary condition).

$$c_e/c_o = 1 - \sum_{m=1}^{\infty} \frac{2\,\beta_m \sin(\beta_m)\exp[\frac{P}{2} - \frac{PT}{4R} - \frac{\beta_m^2 T}{PR}]}{\beta_m^2 + \frac{P^2}{4} + \frac{P}{2}} \qquad [B4]$$

where the eigenvalues β_m are the positive roots of

$$\beta_m \cot(\beta_m) + \frac{P}{2} = 0 \qquad [B5]$$

For $P > 5 + 40\,T$ or $P > 100$, the following approximation is used

$$c_e/c_o = \frac{1}{2}\,\mathrm{erfc}[(\frac{P}{4RT})^{1/2}(R-T)] - (\frac{PT}{\pi R})^{1/2}\exp[-\frac{P}{4RT}(R-T)^2]$$

$$+ \frac{1}{2} (3 + P + \frac{PT}{R}) \exp(P) \operatorname{erfc}[(\frac{P}{4RT})^{1/2} (R+T)] \qquad [B6]$$

Case FN-2 (Finite system, third-type boundary condition).

$$c_e/c_o = 1 - \sum_{m=1}^{\infty} \frac{2 \beta_m \sin(\beta_m) \exp[\frac{P}{2} - \frac{PT}{4R} - \frac{\beta_m^2 T}{PR}]}{\beta_m^2 + \frac{P^2}{4} + P} \qquad [B7]$$

where the eigenvalues β_m are the positive roots of

$$P \beta_m \cot (\beta_m) - \beta_m^2 + \frac{P^2}{4} = 0 \qquad [B8]$$

The approximate solution for $P > 5 + 40T$ or $P > 100$ is

$$c_e/c_o = \frac{1}{2} \operatorname{erfc}[(\frac{P}{4RT})^{1/2} (R-T)]$$

$$+ (\frac{PT}{\pi R})^{1/2} (3 + \frac{P}{2} + \frac{PT}{2R}) \exp[- \frac{P}{4RT} (R-T)^2]$$

$$- \frac{1}{2} [1 + 3P + \frac{4PT}{R} + \frac{P^2}{2R^2} (R+T)^2] \exp(P) \operatorname{erfc}[(\frac{P}{4RT})^{1/2} (R+T)]$$

[B9]

APPENDIX C. CFITM, a computer program for calculating transport parameters from observed solute effluent curves.

This appendix gives a brief description and listing of CFITM, a computer program for calculating the Peclet number (P), the retardation factor (R) and, if needed, the dimensionless pulse time (T_1) from observed effluent data. The program does this by means of a least-squares fit of any of the five analytical expressions in Appendix B to observed effluent data.

The program consists of a main program (MAIN), four subroutines (MODEL, CONC, EIGEN, and MATINV), and one function (EXF). Most of the calculations for the least-squares analysis are carried out in MAIN, including input and output instructions, calculation of a correlation matrix for the different coefficients, and calculation of a 95% confidence interval for each coefficient. Subroutine MODEL calculates the exit concentration for each of the five models given in Appendix B. The choice of the model is governed by the variable MODE: MODE = 0 for Case INF, 1 for SI-1, 2 for SI-2, 3 for FN-1, and 4 for Case FN-2. The analytical solution for Cases FN-1 and FN-2 (MODE = 3,4) requires the evaluation of series (see Appendix B). These calculations are carried out in subroutine CONC. The approximate solutions for FN-1 and FN-2 are also evaluated in CONC. The eigenvalues, β_m, needed for the series solutions, are calculated in subroutine EIGEN. Subroutine MATINV gives a matrix inversion scheme needed for the least-squares analysis in MAIN. The function EXF, finally, is used to calculate the complementary error function (erfc), the exponential function (exp), or the product of erfc and exp.

Table C1 gives a list of the most significant program variables. Table C2 gives instructions regarding set-up of the data cards. The actual input data for example 2 (Chromium transport through sand) are shown in Table 3. The computer output for Example 2 is given in Table C4, while the listing of the program is given in Table C5.

An extra comment is needed for the vector B(I) in Table C1. This vector contains the estimated values of the three coefficients P, R, and T_1 (in that order). If a coefficient is known and a two-parameter curve-fitting is carried out, the measured value of that coefficient should be entered on the fifth data card (see Tables C2 and C3). For a continuous tracer application (no pulse of solute), a large dummy value should be assigned to T_1. This value must exceed all measured pore volumes, Y(I). For Example 2 a dummy value of 100 was assigned to T_1 (card 5 in Table C3). Actually, any value higher than 2.463 could have been used (see card 21 in Table C3).

Table C1. List of the most significant variables in CFITM.

VARIABLE	DEFINITION
B(I)	Vector containing estimates of the coefficients (P, R, T_1).
BI(I)	Vector of coefficient names (P, R, T_1).
EXF(A,B)	Function to calculate exp(A) erfc(B).
G(I)	Eigenvalues (β_m) for Cases FN-1 and FN-2 (MODE = 3,4).
INDEX(I)	Index for each coefficient. If INDEX(I) = 0, the coefficient is known. If INDEX(I) = 1, the coefficient is assumed to be unknown and fitted to the data. For example, if INDEX(1) = 1, INDEX(2) = 0, and INDEX(3) = 1, the coefficients P and T_1 are fitted to the data, while R is assumed to be known (for example from batch equilibrium studies). At least two coefficients need to be unknown.
MIT	Maximum number of iterations in least-squares analysis.
MODE	Model number for the five analytical solutions: MODE = 0 for case INF (infinite system , MODE = 1 for Case SI-1 (semi-finite system, first-type surface boundary condition), MODE = 2 for Case SI-2 (semi-infinite system, third-type boundary conditions), MODE = 3 for Case FN-1 (finite system, first-type boundary condition), and MODE = 4 for Case FN-2 (finite system, third-type boundary condition).
NC	Number of cases considered.
NDATA	Data input code: if NDATA = 1, new data are read in, if NDATA = 0, the same data (or a part of them) are used for the new case. This code allows one to fit the same data to two different models (see Tables C3 and C4).
NIT	Iteration number during least-squares analysis.
NOB	Number of observations (must not exceed 90 with presently dimensioned arrays).
SSQ, SUMB	Residual sum of squares.
STOPCR	Stop criterion: The iterative curve-fitting process stops when the relative change in the ratio of all coefficients becomes less than STOPCR.
TITLE	Vector containing information of title card (input label).
X(I), Y(I)	Observed effluent data (pore volume and concentration, respectively).

Table C2. Data input instructions.

CARD	COLUMNS	FORMAT	VARIABLE	COMMENTS
1	1-5	I5	NC	Number of cases considered.
				The remaining cards are read in for each case. If NDATA = 0 on card 2, data cards 4 through 7 are not needed for that particular case.
2	1-5	I5	MODE	Model number.
2	5-10	I5	NDATA	Data input code.
2	11-15	I5	MIT	Maximum number of iterations.
2	16-20	I5	NOB	Number of observations.
3	1-80	20A4	TITLE	Information card.
4	1-6	A4,A2	BI(1)	Coefficient name for P.
4	11-16	A4,A2	BI(2)	Coefficient name for R.
4	21-26	A4,A2	BI(3)	Coefficient name for T_1.
5	1-10	F10.0	B(1)	Initial value of P.
5	11-20	F10.0	B(2)	Initial value of R.
5	21-30	F10.0	B(3)	Initial value of T_1.
6	1-5	I5	INDEX(1)	Index for each coefficient.
6	6-10	I5	INDEX(2)	See Table C1 for explana-
6	11-15	I5	INDEX(3)	ion.
7, etc.	1-10	F10.0	X(I)	Value of observed pore volume.
7, etc.	11-20	F10.0	Y(I)	Value of observed concentration. Card 7 is repeated NOB times.

Table C3. Data input for example 2.

```
         0         1         2         3         4         5
Column:  12345678901234567890123456789012345678901234567890
Card

  1          2
  2          2    1   15   15
  3                EXAMPLE 2A:  CHROMIUM (COLUMN NUMBER 4)
  4      PECLET     RF       PULSE
  5          20.0       1.30      100.0
  6          1    1    0
  7           0.558     0.000
  8           0.695     0.006
  9           0.831     0.061
 10           0.967     0.198
 11           1.103     0.325
 12           1.239     0.450
 13           1.375     0.592
 14           1.511     0.705
 15           1.647     0.768
 16           1.783     0.841
 17           1.919     0.881
 18           2.055     0.944
 19           2.191     0.966
 20           2.327     0.994
 21           2.463     0.999
 22          4    0   15   15
 23                EXAMPLE 2B:  CHROMIUM (COLUMN NUMBER 4)
```

Table C4. Output for example 2.

```
*************************************************************************
*                                                                       *
*        NON-LINEAR LEAST SQUARES ANALYSIS                              *
*                                                                       *
*        SEMI-INFINITE PROFILE, 3-TYPE BC                               *
*        EXAMPLE 2A CHROMIUM (COLUMN NUMBER 4)                          *
*                                                                       *
*************************************************************************

INITIAL VALUES OF COEFFICIENTS
==============================
 NO      NAME          INITIAL VALUE
  1     PECLET           20.000
  2       RF              1.300
  3     PULSE           100.000

OBSERVED CATA
=============
OBS. NO.      PORE VOLUME      CONCENTRATION
    1           0.5580            0.0
    2           0.6950            0.0060
    3           0.8310            0.0610
    4           0.9670            0.1980
    5           1.1030            0.3250
    6           1.2390            0.4500
    7           1.3750            0.5920
    8           1.5110            0.7050
    9           1.6470            0.7680
   10           1.7830            0.8410
   11           1.9190            0.8810
   12           2.0550            0.9440
   13           2.1910            0.9660
   14           2.3270            0.9940
   15           2.4630            0.9990

ITERATION       SSQ            PECLET          RF
    0        0.0047356        20.00000       1.30000
    1        0.0029914        19.03803       1.28139
    2        0.0029844        19.19318       1.28138
    3        0.0029844        19.18872       1.28137

CORRELATICN MATRIX
==================
          1           2
  1    1.0000
  2    0.2612      1.0000
```

NON-LINEAR LEAST SQUARES ANALYSIS, FINAL RESULTS

VARIABLE	NAME	VALUE	S.E.COEFF.	T-VALUE	95% CONFIDENCE LIMITS LOWER	UPPER
1	PECLET	19.18872	0.9535	20.12	17.1287	21.2488
2	RF	1.28137	0.0068	187.38	1.2666	1.2961

ORDERED BY COMPUTER INPUT					ORDERED BY RESIDUALS				
NO	PORE VOLUME	CONCENTRATION OBS.	CONCENTRATION FITTED	RESI-DUAL	NO	PORE VOLUME	CONCENTRATION OBS.	CONCENTRATION FITTED	RESI-DUAL
1	0.558	0.0	0.003	-0.003	14	2.327	0.994	0.972	0.022
2	0.695	0.006	0.024	-0.018	15	2.463	0.999	0.982	0.017
3	0.831	0.061	0.082	-0.021	4	0.967	0.198	0.183	0.015
4	0.967	0.198	0.183	0.015	5	1.103	0.325	0.314	0.011
5	1.103	0.325	0.314	0.011	12	2.055	0.944	0.934	0.010
6	1.239	0.450	0.455	-0.005	13	2.191	0.966	0.957	0.009
7	1.375	0.592	0.586	0.006	8	1.511	0.705	0.697	0.008
8	1.511	0.705	0.697	0.008	7	1.375	0.592	0.586	0.006
9	1.647	0.768	0.786	-0.018	1	0.558	0.0	0.003	-0.003
10	1.783	0.841	0.852	-0.011	6	1.239	0.450	0.455	-0.005
11	1.919	0.881	0.900	-0.019	10	1.783	0.841	0.852	-0.011
12	2.055	0.944	0.934	0.010	9	1.647	0.768	0.786	-0.018
13	2.191	0.966	0.957	0.009	2	0.695	0.006	0.024	-0.018
14	2.327	0.994	0.972	0.022	11	1.919	0.881	0.900	-0.019
15	2.463	0.999	0.982	0.017	3	0.831	0.061	0.082	-0.021

END OF PROBLEM

```
****************************************************************
*                                                              *
*        NCN-LINEAR LEAST SQUARES ANALYSIS                     *
*                                                              *
*        FINITE PROFILE, 3-TYPE BC                             *
*        EXAMPLE 2B  CHROMIUM  (COLUMN NUMBER 4)               *
*                                                              *
****************************************************************
```

OBSERVED CATA
=============

OBS. NO.	PORE VOLUME	CONCENTRATION
1	0.5580	0.0
2	0.6950	0.0060
3	0.8310	0.0610
4	0.9670	0.1980
5	1.1030	0.3250
6	1.2390	0.4500
7	1.3750	0.5920
8	1.5110	0.7050
9	1.6470	0.7680
10	1.7830	0.8410
11	1.9190	0.8810
12	2.0550	0.9440
13	2.1910	0.9660
14	2.3270	0.9940
15	2.4630	0.9990

ITERATION	SSQ	PECLET	RF
0	0.0141245	20.00000	1.30000
1	0.0030266	18.28417	1.34697
2	0.0029795	18.60950	1.34848
3	0.0029795	18.58894	1.34851
4	0.0029795	18.59020	1.34851

CORRELATION MATRIX
==================

	1	2
1	1.0000	
2	-0.2061	1.0000

NON-LINEAR LEAST SQUARES ANALYSIS, FINAL RESULTS

VARIABLE	NAME	VALUE	S.E.COEFF.	T-VALUE	95% CONFIDENCE LIMITS LOWER	95% CONFIDENCE LIMITS UPPER
1	PECLET	18.59020	0.9509	19.55	16.5359	20.6445
2	RF	1.34851	0.0071	190.08	1.3332	1.3638

---ORDERED BY COMPUTER INPUT---

NO	PORE VOLUME	CONCENTRATION OBS.	CONCENTRATION FITTED	RESI-DUAL
1	0.558	0.0	0.003	-0.003
2	0.695	0.006	0.024	-0.018
3	0.831	0.061	0.082	-0.021
4	0.967	0.198	0.183	0.015
5	1.103	0.325	0.315	0.010
6	1.239	0.450	0.455	-0.005
7	1.375	0.592	0.586	0.006
8	1.511	0.705	0.697	0.008
9	1.647	0.768	0.786	-0.018
10	1.783	0.841	0.852	-0.011
11	1.919	0.881	0.900	-0.019
12	2.055	0.944	0.933	0.011
13	2.191	0.966	0.956	0.010
14	2.327	0.994	0.972	0.022
15	2.463	0.999	0.982	0.017

---ORDERED BY RESIDUALS---

NO	PORE VOLUME	CONCENTRATION OBS.	CONCENTRATION FITTED	RESI-DUAL
14	2.327	0.994	0.972	0.022
15	2.463	0.999	0.982	0.017
4	0.967	0.198	0.183	0.015
12	2.055	0.944	0.933	0.011
5	1.103	0.325	0.315	0.010
13	2.191	0.966	0.956	0.010
8	1.511	0.705	0.697	0.008
7	1.375	0.592	0.586	0.006
1	0.558	0.0	0.003	-0.003
6	1.239	0.450	0.455	-0.005
10	1.783	0.841	0.852	-0.011
9	1.647	0.768	0.786	-0.018
2	0.695	0.006	0.024	-0.018
11	1.919	0.881	0.900	-0.019
3	0.831	0.061	0.082	-0.021

END OF PROBLEM

TABLE C5. Fortran listing of CFITM.

MAIN

```
C
C
C      ******************************************************************
C      *                                                                *
C      *        NON-LINEAR LEAST-SQUARES ANALYSIS              CFITM    *
C      *                                                                *
C      *                                                                *
C      *        -------------- INPUT INFORMATION --------------         *
C      *                                                                *
C      *        CARD 1: NUMBER OF CASES CONSIDERED: NC (I5)             *
C      *                                                                *
C      *                THE NEXT CARDS ARE REPEATED NC TIMES            *
C      *                                                                *
C      *        CARD 2: MODEL  NUMBER (MODE), DATA INPUT CODE (NDATA),  *
C      *                MAXIMUM NUMBER OF ITERATIONS (MIT) AND NUMBER   *
C      *                OF OBSERVATIONS (NOB)  (4I5)                    *
C      *        CARD 3: INFORMATION CARD  (20A4)                        *
C      *                                                                *
C      *                THE NEXT CARDS ARE READ IN ONLY IF NDATA=1      *
C      *                                                                *
C      *        CARD 4: NAMES OF THE COEFFICIENTS  3(A4,A2,4X)          *
C      *        CARD 5: INITIAL ESTIMATES OF COEFFICIENTS (3F10.0)      *
C      *        CARD 6: INDEX FOR EACH COEFFICIENT  5(I5)               *
C      *                =0  IF COEFFICIENT IS KNOWN (CONSTANT)          *
C      *                =1  IF COEFFICIENT IS UNKNOWN                   *
C      *        CARD 7,ETC.: EXPERIMENTAL DATA: PORE VOLUME AND         *
C      *               CONCENTRATION (NOB CARDS)  (2F10.0)              *
C      *                                                                *
C      ******************************************************************
C
C
       IMPLICIT REAL*8(A-H,O-Z)
       DIMENSION Y(90),X(90),F(90),R(90),DELZ(90,5),B(10),E(5),TH(10),
      1P(5),PHI(5),Q(5),LSORT(90),TB(10),A(5,5),BI(10),TITLE(20),D(5,5),
      2INDEX(5)
       DATA STOPCR/0.0005/
C
C
C      ----- READ NUMBER OF CASES -----
       READ(5,1006) NC
       DO 120 NCASE=1,NC
C
       WRITE(6,1000)
C
C      ----- READ INPUT PARAMETERS -----
       READ(5,1006) MODE,NDATA,MIT,NOB
       IF(MODE.EQ.0) WRITE(6,1021)
       IF(MODE.EQ.1) WRITE(6,1022)
       IF(MODE.EQ.2) WRITE(6,1023)
       IF(MODE.EQ.3) WRITE(6,1024)
       IF(MODE.EQ.4) WRITE(6,1025)
       READ(5,1001) TITLE
       WRITE(6,1002) TITLE
       IF(NDATA.EQ.0) GO TO 10
```

MAIN

```
C
C     ----- REAC COEFFICIENTS NAMES -----
      READ(5,10C4) (BI(I),I=1,6)
C
C     ----- REAC INITIAL ESTIMATES -----
      READ(5,10C5) (B(I),I=4,6)
C
C     ----- REAC INDICES -----
      READ(5,10C6) (INDEX(I),I=1,3)
      WRITE(6,1007)
      DO 4  I=1,3
      J=2*I-1
    4 WRITE(6,1008) I,BI(J),BI(J+1),B(I+3)
C
C     ----- REAC AND WRITE EXPERIMENTAL DATA -----
      DC 6 I=1,NOB
    6 REAC(5,10C5) X(I),Y(I)
   10 WRITE(6,1009)
      DO 12 I=1,NCB
   12 WRITE(6,1C10) I,X(I),Y(I)
C
C     ----------
      NP=0
      DO 14 I=4,6
      TB(I)=B(I)
      IF(INDEX(I-3).EQ.0) GO TO 14
      NP=NP+1
      K=2*NP-1
      J=2*I-7
      BI(K)=BI(J)
      BI(K+1)=BI(J+1)
      B(NP)=B(I)
      TH(NP)=B(NP)
   14 TH(I)=B(I)
C
C     ----------
      GA=0.02
      NIT=0
      NP2=2*NP
      CALL MODEL(TH,F,NOB,X,INDEX,MODE)
      SSQ=0.
      DO 32 I=1,NOB
      R(I)=Y(I)-F(I)
   32 SSQ=SSQ+R(I)*R(I)
      WRITE(6,1C11) (BI(J),BI(J+1),J=1,NP2,2)
      WRITE(6,1012) NIT,SSQ,(B(I),I=1,NP)
C
C     ----- BEGIN OF ITERATION -----
   34 NIT=NIT+1
      GA=0.1*GA
      DO 38 J=1,NP
      TEMP=TH(J)
      TH(J)=1.01*TH(J)
      Q(J)=0
```

MAIN

```
      CALL MODEL(TH,DELZ(1,J),NOB,X,INDEX,MODE)
      DO 36 I=1,NOB
      DELZ(I,J)=DELZ(I,J)-F(I)
   36 Q(J)=Q(J)+DELZ(I,J)*R(I)
      Q(J)=100.*Q(J)/TH(J)
C
C     ----- Q=XT*R (STEEPEST DESCENT) -----
   38 TH(J)=TEMP
      DO 44 I=1,NP
      DO 42 J=1,I
      SUM=0
      DO 40 K=1,NOB
   40 SUM=SUM+DELZ(K,I)*DELZ(K,J)
      D(I,J)=10000.*SUM/(TH(I)*TH(J))
   42 D(J,I)=D(I,J)
   44 E(I)=DSQRT(D(I,I))
   50 DO 52 I=1,NP
      DO 52 J=1,NP
   52 A(I,J)=D(I,J)/(E(I)*E(J))
C
C     ----- A IS THE SCALED MOMENT MATRIX -----
      DO 54 I=1,NP
      P(I)=Q(I)/E(I)
      PHI(I)=P(I)
   54 A(I,I)=A(I,I)+GA
      CALL MATINV(A,NP,P)
C
C     ----- P/E IS THE CORRECTION VECTOR -----
      STEP=1.0
   56 DO 58 I=1,NP
   58 TB(I)=P(I)*STEP/E(I)+TH(I)
      DO 62 I=1,NP
      IF(TH(I)*TB(I))66,66,62
   62 CONTINUE
      SUMB=0
      CALL MODEL(TB,F,NOB,X,INDEX,MODE)
      DO 64 I=1,NOB
      R(I)=Y(I)-F(I)
   64 SUMB=SUMB+R(I)*R(I)
   66 SUM1=0.0
      SUM2=0.0
      SUM3=0.0
      DO 68 I=1,NP
      SUM1=SUM1+P(I)*PHI(I)
      SUM2=SUM2+P(I)*P(I)
   68 SUM3=SUM3+PHI(I)*PHI(I)
      ARG=SUM1/DSQRT(SUM2*SUM3)
      ANGLE=57.29578*DATAN2(DSQRT(1.-ARG*ARG),ARG)
C
C     ----------
      DO 72 I=1,NP
      IF(TH(I)*TB(I))74,74,72
   72 CONTINUE
      IF(SUMB/SSQ-1.0)80,80,74
   74 IF(ANGLE-30.0)76,76,78
```

MAIN

```
   76 STEP=0.5*STEP
      GO TO 56
   78 GA=10.*GA
      GO TO 50
C
C     ----- PRINT COEFFICIENTS AFTER EACH ITERATION -----
   80 CONTINUE
      DO 82 I=1,NP
   82 TH(I)=TB(I)
      WRITE(6,1012) NIT,SUMB,(TH(I),I=1,NP)
      DO 86 I=1,NP
      IF(DABS(P(I)*STEP/E(I))/(1.0D-20+DABS(TH(I)))-STOPCR) 86,86,94
   86 CONTINUE
      GO TO 96
   94 SSQ=SUMB
      IF(NIT.LE.MIT) GO TO 34
C
C     ----- END OF ITERATION LOOP -----
   96 CONTINUE
      CALL MATINV(D,NP,P)
C
C     ----- WRITE CORRELATION MATRIX -----
      DO 98 I=1,NP
   98 E(I)=DSQRT(D(I,I))
      WRITE(6,1013) (I,I=1,NP)
      DO 102 I=1,NP
      DO 100 J=1,I
  100 A(J,I)=D(J,I)/(E(I)*E(J))
  102 WRITE(6,1014) I,(A(J,I),J=1,I)
C
C     ----- CALCULATE 95% CONFIDENCE INTERVAL -----
      Z=1./FLOAT(NOB-NP)
      SDEV=DSQRT(Z*SUMB)
      TVAR=1.96+Z*(2.3779+Z*(2.7135+Z*(3.187936+2.466666*Z**2)))
      WRITE(6,1015)
      DO 108 I=1,NP
      SECOEF=E(I)*SDEV
      TVALUE=TH(I)/SECOEF
      TSEC=TVAR*SECOEF
      TMCOE=TH(I)-TSEC
      TPCOE=TH(I)+TSEC
      J=2*I-1
  108 WRITE(6,1016) I,BI(J),BI(J+1),TH(I),SECOEF,TVALUE,TMCOE,TPCOE
C
C     ----- PREPARE FINAL OUTPUT -----
      LSORT(1)=1
      DO 116 J=2,NOB
      TEMP=R(J)
      K=J-1
      DO 111 L=1,K
      LL=LSORT(L)
      IF(TEMP-R(LL)) 112,112,111
  111 CONTINUE
      LSORT(J)=J
```

MAIN

```
      GO TO 116
  112 KK=J
  113 KK=KK-1
      LSORT(KK+1)=LSORT(KK)
      IF(KK-L) 115,115,113
  115 LSORT(L)=J
  116 CONTINUE
      WRITE(6,1017)
      DO 118 I=1,NOB
      J=LSORT(NOB+1-I)
  118 WRITE(6,1018) I,X(I),Y(I),F(I),R(I),J,X(J),Y(J),F(J),R(J)
  120 WRITE(6,1020)
C
C     ----- END OF PROBLEM -----
 1000 FORMAT(1H1,10X,82(1H*)/11X,1H*,80X,1H*/11X,1H*,10X,'NON-LINEAR LEA
     1ST SQUARES ANALYSIS',37X,1H*/11X,1H*,80X,1H*)
 1001 FORMAT(20A4)
 1002 FORMAT(11X,1H*,20A4,1H*/11X,1H*,80X,1H*/11X,82(1H*))
 1004 FORMAT(5(A4,A2,4X))
 1005 FORMAT(5F10.0)
 1006 FORMAT(5I5)
 1007 FORMAT(//11X,'INITIAL VALUES OF COEFFICIENTS'/11X,30(1H=)/12X,'NO'
     1,6X,'NAME',9X,'INITIAL VALUE')
 1008 FORMAT(11X,I3,5X,A4,A2,4X,F12.3)
 1009 FORMAT(//11X,'OBSERVED DATA',/11X,13(1H=)/11X,'OBS. NO.',5X,'PORE
     1VOLUME',5X,'CONCENTRATION')
 1010 FORMAT(11X,I5,5X,F12.4,4X,F12.4)
 1011 FORMAT(//11X,'ITERATION',6X,'SSQ',4X,5(7X,A4,A2))
 1012 FORMAT(11X,I5,5X,F11.7,2X,5F13.5)
 1013 FORMAT(///,11X,'CORRELATION MATRIX'/11X,18(1H=)/14X,10(4X,I2,5X))
 1014 FORMAT(11X,I3,10(2X,F7.4,2X))
 1015 FORMAT(1H1,10X,'NON-LINEAR LEAST SQUARES ANALYSIS, FINAL RESULTS'
     1/11X,48(1H=)//72X,'95% CONFIDENCE LIMITS'/11X,'VARIABLE',4X,'NAME'
     2,8X,'VALUE',8X,'S.E.COEFF.',3X,'T-VALUE',5X,'LOWER',10X,'UPPER')
 1016 FORMAT(14X,I2,6X,A4,A2,2X,F12.5,5X,F9.4,4X,F8.2,2X,F9.4,6X,F9.4)
 1017 FORMAT(//10X,9(1H-),'ORDERED BY COMPUTER INPUT',10(1H-), 7X,12(1H-
     1),'ORDERED BY RESIDUALS',12(1H-)/18X,'PORE',6X,'CONCENTRATION',
     26X,'RESI-',18X,'PORE',6X,'CONCENTRATION',6X,'RESI-'/10X,'NO',4X,
     3'VOLUME',6X,'OBS.',4X,'FITTED',6X,'DUAL',10X,'NO',4X,'VOLUME',6X,
     4'OBS.',4X,'FITTED',6X,'DUAL')
 1018 FORMAT(10X,I2,4F10.3,10X,I2,4F10.3)
 1020 FORMAT(///11X,'END OF PROBLEM'/11X,14(1H=))
 1021 FORMAT(11X,1H*,10X,'INFINITE PROFILE',54X,1H*)
 1022 FORMAT(11X,1H*,10X,'SEMI-INFINITE PROFILE, 1-TYPE BC',38X,1H*)
 1023 FORMAT(11X,1H*,10X,'SEMI-INFINITE PROFILE, 3-TYPE BC',38X,1H*)
 1024 FORMAT(11X,1H*,10X,'FINITE PROFILE, 1-TYPE BC',45X,1H*)
 1025 FORMAT(11X,1H*,10X,'FINITE PROFILE, 3-TYPE BC',45X,1H*)
      STOP
      END
```

EIGEN

```
      SUBROUTINE EIGEN(G,P,MODE)
C
C     PURPOSE: TO CALCULATE THE EIGENVALUES
C
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION G(20)
      BETA=0.1
      S=0.0
      IF(MODE.EQ.4) S=1.0
      DO 4 I=1,20
      J=0
    1 J=J+1
      IF(J.GT.15) GO TO 3
      DELTA=-0.2*(-0.5)**J
    2 BET2=BETA
      BETA=BETA+DELTA
      A=BET2*DCOS(BET2)+(0.25*(2.-S)*P-S*BET2**2/P)*DSIN(BET2)
      B=BETA*DCOS(BETA)+(0.25*(2.-S)*P-S*BETA**2/P)*DSIN(BETA)
      IF(A*B)1,3,2
    3 G(I)=(BET2*B-BETA*A)/(B-A)
    4 BETA=BETA+0.2
      RETURN
      END
```

CONC

```
      SUBROUTINE CONC(C,G,P,T,MODE)
C
C     PURPOSE: TO CALCULATE CONCENTRATION C FOR MODE=3,4
C
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION G(20)
      E=0.0
      TOL=.00001
      S=DMIN1(1.D02,5.+40.*T)
      IF(P.GE.S) GO TO 4
      S=1
      IF(MODE.EQ.3) S=0.5
C
C     -----SERIES SOLUTION-----
      EX=0.5*P-0.25*P*T
      SUM=0.0
      DO 2 J=1,10
      DSUM=0.0
      DO 1 K=1,2
      I=2*J+K-2
      A=G(I)*DSIN(G(I))
      IF(DABS(A).LT.1.D-04) A=0.0
      EXP=EX-G(I)**2*T/P
    1 DSUM=DSUM+A*EXF(EXP,E)/(G(I)**2+0.25*P*P+S*P)
      SUM=SUM+DSUM
      IF(DABS(DSUM/SUM).LT.TOL) GO TO 3
    2 CONTINUE
      GO TO 4
    3 C=1.-2.*SUM
      RETURN
    4 AM=0.5*(1.-T)*DSQRT(P/T)
      AP=0.5*(1.+T)*DSQRT(P/T)
      A=0.5*EXF(E,AM)
      B=0.5*EXF(P,AP)
      D=DSQRT(.3183099*P*T)*EXF(-AM*AM,E)
      IF(MODE.EQ.3) C=A+(3.+P+P*T)*B-D
      IF(MODE.EQ.4) C=A+(3.+.5*P+.5*P*T)*D-(1.+3.*P+P*T*(4.+2.*AP**2))*B
      RETURN
      END
```

EXF

```
      FUNCTION EXF(A,B)
C
C     PURPOSE: TO CALCULATE EXP(A) ERFC(B)
C
      IMPLICIT REAL*8 (A-H,O-Z)
      EXF=0.0
      IF((DABS(A).GT.170.).AND.(B.LE.0.)) RETURN
      C=A-B*B
      IF((DABS(C).GT.170.).AND.(B.GE.0.)) RETURN
      IF(C.LT.-170.) GO TO 3
      X=DABS(B)
      IF(X.GT.3.0) GO TO 1
      T=1./(1.+.3275911*X)
      Y=T*(.2548296-T*(.2844967-T*(1.421414-T*(1.453152-1.061405*T))))
      GO TO 2
    1 Y=.5641896/(X+.5/(X+1./(X+1.5/(X+2./(X+2.5/X+1.)))))
    2 EXF=Y*DEXP(C)
    3 IF(B.LT.0.0) EXF=2.*DEXP(A)-EXF
      RETURN
      END
```

MATINV

```
      SUBROUTINE MATINV(A,NP,B)
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION A(5,5),B(10),INDEX(5,2)
      DO 2 J=1,5
    2 INDEX(J,1)=0
      I=0
    4 AMAX=-1.0
      DO 12 J=1,NP
      IF(INDEX(J,1)) 12,6,12
    6 DO 10 K=1,NP
      IF(INDEX(K,1)) 10,8,10
    8 P=DABS(A(J,K))
      IF(P.LE.AMAX) GO TO 10
      IR=J
      IC=K
      AMAX=P
   10 CONTINUE
   12 CONTINUE
      IF(AMAX) 30,30,14
   14 INDEX(IC,1)=IR
      IF(IR.EQ.IC) GO TO 18
      DO 16 L=1,NP
      P=A(IR,L)
      A(IR,L)=A(IC,L)
   16 A(IC,L)=P
      P=B(IR)
      B(IR)=B(IC)
      B(IC)=P
      I=I+1
      INDEX(I,2)=IC
   18 P=1./A(IC,IC)
      A(IC,IC)=1.0
      DO 20 L=1,NP
   20 A(IC,L)=A(IC,L)*P
      B(IC)=B(IC)*P
      DO 24 K=1,NP
      IF(K.EQ.IC) GO TO 24
      P=A(K,IC)
      A(K,IC)=0.0
      DO 22 L=1,NP
   22 A(K,L)=A(K,L)-A(IC,L)*P
      B(K)=B(K)-B(IC)*P
   24 CONTINUE
      GO TO 4
   26 IC=INDEX(I,2)
      IR=INDEX(IC,1)
      DO 28 K=1,NP
      P=A(K,IR)
      A(K,IR)=A(K,IC)
   28 A(K,IC)=P
      I=I-1
   30 IF(I) 26,32,26
   32 RETURN
      END
```

MODEL

```
      SUBROUTINE MODEL(B,Y,NOB,X,INDEX,MODE)
C
C     PURPOSE: TO CALCULATE CONCENTRATIONS FOR GIVEN PORE VOLUME
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION B(10),Y(90),X(90),INDEX(5),G(20)
      E=0.
      K=0
      DO 2 I=4,6
      IF(INDEX(I-3).EQ.0) GO TO 2
      K=K+1
      B(I)=B(K)
    2 CONTINUE
      P=B(4)
      R=B(5)
      IF((P.LE.100.).AND.(MODE.GE.3)) CALL EIGEN (G,P,MODE)
      DO 6 J=1,NOB
      DO 4 M=1,2
      C=0.0
      T=(X(J)+(1-M)*B(6))/R
      IF(T.LE.C.) GO TO 6
      AM=0.5*(1.-T)*DSQRT(P/T)
      AP=0.5*(1.+T)*DSQRT(P/T)
      IF(MODE.EQ.0) C=0.5*EXF(E,AM)
      IF(MODE.EQ.1) C=0.5*EXF(E,AM)+0.5*EXF(P,AP)
      IF(MODE.EQ.2) C=0.5*EXF(E,AM)+DSQRT(.3183099*P*T)*EXF(-AM*AM,E)-
     10.5*(1.+P+P*T)*EXF(P,AP)
      IF(MODE.GE.3) CALL CONC(C,G,P,T,MODE)
      IF(M.EQ.2) GO TO 6
      Y(J)=C
    4 CONTINUE
    6 Y(J)=Y(J)-C
      RETURN
      END
```